

Creating an Interactive Excel Chart

By Neale Blackwood

www.a4accounting.com.au

a4@iinet.net.au



A4 ACCOUNTING
EXCEL CONSULTING, TRAINING AND WEBINARS

Introduction

This session is based on an article I wrote in the August 2012 issue of the INTHEBLACK magazine and it was also the first webinar that I ran in August 2012. This manual provides more detail on the techniques involved.

Charts convert tables of values into a graphic representation of those values. Done well, charts can convey trends and relationships in a glance. An interactive chart provides users with a simple interface to control what is displayed on the chart. Below is an example of an interactive chart that we will create in this session.

Excel charts are typically static. They display the same measures and change periodically as new data is added.

Static charts are commonly used in monthly reports and dashboards. Changing a static chart requires manual updating of the chart's data source.



The example we will work with takes advantage of two of Excel's Form controls that allow the user to easily change the values plotted on the chart.

Form controls provide a structured, easy to use interface for users to make selections. They are faster than typing entries into a cell and can improve the user interface for your spreadsheets.

Interactive charts are most commonly used for those times when you need flexibility. Analysis work often requires handling large data tables. The data may need to be compared or relationships identified between data tables. This may require many reports and many separate charts.

Interactive charts take more time to set up, but once created provide an analysis tool that allows investigation of trends and relationships without the requirement of creating multiple charts.

The techniques involved in creating interactive charts can be applied to other areas in Excel such as report creation. Indeed, the controls are used to create flexible reports that become the data source for the interactive chart.

Versions

The following instructions relate to Excel 2016. Everything in this article can be done in all versions, but the screen shots, options and formula may differ.

Scenario

The table below shows financial and non-financial measures. This table extends to cover 24 months.

Measures	Jul-10	Aug-10	Sep-10	Oct-10	Nov-10	Dec-10	Jan-11
Sales \$	\$85,275	\$78,002	\$79,945	\$96,499	\$82,673	\$76,953	\$96,884
Margin \$	\$52,018	\$53,041	\$51,165	\$64,654	\$45,470	\$50,019	\$54,255
Margin %	61%	68%	64%	67%	55%	65%	56%
Profit \$	\$21,319	\$22,621	\$27,181	\$24,125	\$28,936	\$26,164	\$31,972
Units sold	10,659	8,667	11,421	13,786	9,186	15,391	13,841
Head count	50	51	48	52	52	53	54
Customer satisfaction %	74%	80%	81%	82%	78%	79%	75%

The goal is to compare financial and non-financial measures over time, to find any relationships.

The first step in creating an interactive chart is to identify what the user needs to change on the chart.

To achieve our goal the user will need to manipulate three parameters:

1. **Financial measures** – choose between sales, margin, margin percentage and profit
2. **Non-financial measures** – choose between units sold, head count and customer satisfaction percentage
3. **Number of periods to analyse** – select a number between 6 to 24. I have used 6 as the minimum to provide sufficient data points to identify trends and relationships.

The spreadsheet controls that we will use are

- **Combo Box** – drop down selection of the financial and non-financial measures to compare
- **Scroll Bar** – selection to vary the number of months to include in the chart

We will work through the Blank version of the Excel file to create the final result. It has much of the structure in place but needs controls and formulas.

Form Controls

To use Form controls you need to make the Developer Ribbon tab visible.

Excel 2007 – click the round Office button (top left corner of the screen) and select Excel Options. Tick the Show Developer tab in the Ribbon option and click OK.

Excel 2010 and later – right-click the Ribbon and choose Customize the Ribbon. Tick the Developer tab check box on the right side of the dialog and click OK.

On the Developer tab the Insert icon drop down displays the Form Controls and ActiveX Controls. Pointing to a control's icon will display the control's name.



Active X Controls - these are similar to Form Controls but are more complex to use and are not covered in this article.

Form Controls are the simplest to use and are less likely to cause errors.

Creating the Interactive Chart


The Controls

Combo Box

This control displays a drop down list. A Combo Box requires a range that contains the entries to be displayed in the drop down list. We require two Combo Boxes. One each for Financial and Non-Financial measures. I have created two lists in a sheet called Lists.

	A	B	C
1	Financial		Non_Financial
2	Sales \$		Units sold
3	Margin \$		Head count
4	Margin %		Customer satisfaction %
5	Profit \$		

It is best practice to create range names for lists used with controls. We will create two range names in the Lists sheet to work with the Combo Box controls. Excel has a shortcut that automatically creates range names based on existing entries.




In the Lists sheet select the range A1:A5 as per image above. Click the Formulas Ribbon tab and the click the Create from Selection icon.

A dialog will display. Ensure it has the Top row option ticked and click OK. This creates range names based on cell labels, in this case the top row.

The name created is "Financial" based on cell A1. The named range will be A2:A5.

To create the non-financial list range name, select the range C1:C4 and press the F4 key, which repeats the last action and this will create another range name called "Non_Financial" based on cell C1 with the named range being C2:C4.





There is a separate Chart sheet where the interactive chart will be created. It is best practice when creating Excel files to use separate sheets for data, lists, workings and reports.

In the Chart sheet click the Developer Ribbon tab and click the Insert icon.

Click the Combo Box icon which is the second icon from the left on the top line. Use the mouse to draw the Combo Box on the sheet above cell C8.

Then right click the Combo Box and choose Format Control. Make the following entries as per the image below and Click OK.



The Input range: box is where you define the list to be displayed in the drop down box. "Financial" is the range name defined earlier and lists the financial measures to include in the chart.

The Cell link: box is the cell on the sheet that is updated whenever the control is used. In many cases you would place the control above this linked cell to hide it. As this is a demonstration of controls, having the linked cell visible will assist in understanding how the control works.

The Drop down lines: allows you to control how many lines are displayed when the drop down arrow is clicked.

The Cell link is updated with a number representing the selected option's position in the list. Clicking the second entry in the list will enter 2 in the linked cell. The linked cell is usually placed below the control itself which hides it. I have used a visible cell, next to the control, to demonstrate how the control updates the cell.


To resize or move a control right click it and press the Esc key. You can then click, hold and drag the control with the mouse to move it. When it is selected you can also use the arrow keys to move it small distances. Use the round icons around the edge of the control to re-size it.

The combo box is ready to use now.

	A	B	C
4			
5			Select Chart Parameters
6			
7			Financial
8		4	Profit \$
9			


Create another Combo Box above cell C11 and use these options for the Non-Financial drop down.

The Input range box contains the list range name created earlier.



It is best practice to name cells that are linked to controls. The usual practice is to include the control's name as the first part of the range name. This makes it obvious when reviewing a formula that a range name is associated with a control.

To create the first range name click cell B8 then click inside the Name Box which is located to the left of the Formula bar and above the column letters, see below.




Type `combo_Financial` in the Name Box (as above) and then press Enter. The name will be centred in the Name Box if it is accepted. The underscore character is used because spaces are not permitted in range names.

Click cell B11 and name it `combo_Non_Financial` in the Name Box and press Enter.


Scroll Bar

A Scroll Bar will complete the user interface. A Scroll Bar control allows the user to increase or decrease a number by clicking the mouse in different spots on the Scroll Bar or by clicking and dragging the Scroll Bar "thumb" (the rectangle shape that moves along the Scroll Bar). See image below for the parts of the Scroll Bar. The values mentioned below are based on the settings used on the Format Control dialog on the next page.



The Scroll Bar is the third icon on the second line of the Insert Control list.

Click on the Scroll Bar icon and draw a Scroll Bar above cell C15.




Right click the Scroll Bar and select Format Control. Change the settings as per image.

The *Minimum value* for the Scroll Bar is six which represents six months. The *Maximum value* of 24 represents the limit of the data, which has 24 months of data.

The *Incremental change*: box defines how much the value changes when the arrows on either end of the Scroll Bar are clicked.

The *Page change*: box is the incremental value when you click between the Scroll Bar thumb and the arrow keys on either end.



Each click of an arrow will change the current value in cell C14 by one. Clicking between the arrow and the thumb will change the value by three.

Click cell C14 and click in the Name Box and type scroll_NumMonths and press Enter.

We now have all the entries we require to create the flexible report that will become the data source for the chart.

The Report

	A	B	C	D	E	F	G	H	I	J	K	L	M
1			Number of Months =>	1	2	3	4	5	6	7	8	9	10
2			Show month	FALSE	TRUE	TRUE							
3			Sales \$ and Units sold										
4													
5				#N/A	Mar-11	Apr-11							
6	Financial		Sales \$	#N/A	85655	91779							
7	Non Financial		Units sold	#N/A	12236	11472							
8													
9			Measures	Jul-10	Aug-10	Sep-10	Oct-10	Nov-10	Dec-10	Jan-11	Feb-11	Mar-11	Apr-11
10			Sales \$	85,275	78,002	79,945	96,499	82,673	76,953	96,884	97,700	85,655	91,779
11			Margin \$	52,018	53,041	51,165	64,654	45,470	50,019	54,255	55,689	48,823	55,067
12			Margin %	61%	68%	64%	67%	55%	65%	56%	57%	57%	60%
13			Profit \$	21,319	22,621	27,181	24,125	28,936	26,164	31,972	30,287	23,127	32,123
14			Units sold	10,659	8,667	11,421	13,786	9,186	15,391	13,841	10,856	12,236	11,472
15			Head count	50	51	48	52	52	53	54	50	51	52
16			Customer satisfaction %	74%	80%	81%	82%	78%	79%	75%	74%	76%	77%

The above image shows the Chart Tables sheet. This layout makes it easy to create the report that will become the source data for the chart.

The possible measures to include in the chart are listed in rows 10 to 16. Rows 5 to 7 will become the data source for the chart using INDEX and SUMIF functions. Note: the columns extend across to column AA.

In the report there are seven important formulas marked in green. We will create these formulas, understanding these formulas and how they work together will allow you to create your own flexible reports and charts.

Explanation of the formulas on Chart Table sheet

Row 2 establishes whether the month number should be plotted on the chart based on the number of months selected using the Scroll Bar. The assumption is that we will always plot the last month of the data. The formula in cell D2 is

=D1>(24-scroll_NumMonths)

This is a logical test formula that you usually see in an IF function, but it can be entered into a cell by itself and it will return either TRUE or FALSE.

Subtracting the range name scroll_NumMonth from 24 will provide the first month number to plot on the chart. If the number of months in row 1 is greater than the first month number to plot, then a TRUE will be displayed. Otherwise a FALSE will be displayed. This will be used in later formulas to determine which values to display in the columns. This formula in D2 is copied to all cells in row 2 across to column AA2.

Row 2 is an example of centralising the logic for a column in one row instead of using multiple IF statements within other formulas using the same logical test.

The formula in cell C3 is

```
=C6&" and "&C7
```

This formula creates the text for the chart title and will be used in the chart to create a dynamic chart title. The & symbol combines text entries.

The formula in cell D5 is

```
=IF(D2,D9,NA())
```

This formula is copied across to AA5 and will display the month from row 9 if there is a TRUE in row 2. If row 2 contains FALSE it will display the error message #N/A. The NA() function displays the #N/A error message.

Charts and Errors – Excel charts do not plot error values. The formula in D5 uses that fact to only plot the months that are required. The formula in cell C6 is

```
=IFERROR(INDEX(Financial,combo_Financial),"")
```

The INDEX function extracts the name of the measure selected by the financial Combo Box. The range name "Financial" is in the Lists sheet and holds the entries for the drop down list for the financial Combo Box. The range name combo_Financial is the linked cell for the financial Combo Box and holds a number that represents the current selection from the Combo Box.

The IFERROR function handles the situation where the Combo Box has no selection. The INDEX function would return an error if combo_Financial was blank. The IFERROR function will display a blank cell if that occurs. The two inverted commas "" will display a blank cell.

The formula in cell C7 is

```
=IFERROR(INDEX(Non_Financial,combo_Non_Financial),"")
```

This formula does exactly the same as C6, except it works with the non-financial entries. The formula in D6 is

```
=IF(AND(D$2,$C6<>""),SUMIF($C$10:$C$16,$C6,D$10:D$16),NA())
```

This formula uses an IF function in combination with an AND function to determine the values to plot. The AND function will return TRUE if both D2 is TRUE and C6 is not blank. D2 determines if the month should be plotted. C6 will only be blank if there is no entry in the financial Combo Box. When the AND function returns TRUE the SUMIF function is used to extract the correct values from the table below based on the entry in C6.

This formula also uses the NA() function to display #N/A to stop values being plotted on the chart.


Cell D6 is copied to E6:AA6 and D7:AA7.

The Interactive chart

The last step is to create the chart. We will create it on the Charts Table sheet and then move it to the Chart Sheet where the controls are.

Select the range C5:AA7 in the Chart Table sheet and click the Insert Ribbon tab and click the Line chart icon drop down.

Select the first Line chart, as the per the image.



Depending on your selections in the Combo Boxes, it should look something like the image on the right.

First delete the chart legend (bottom of image) by selecting it and pressing the Delete key on the keyboard.

Colours will be added to the Chart sheet next to the Combo Boxes to provide a legend.




The Non-Financial measures can vary between values and percentages. This means the chart could be plotting dollar values against percentages, which would not be readable as the percentage would be too close to the bottom axis to be readable.

To ensure the chart plots the values correctly we need to plot the Non-Financial data series against a secondary vertical axis.

This is done by right clicking the Non-Financial line (it should be red) and choosing Format Data Series. In the Task Pane on the right choose Secondary Axis – see image on right.

To make it more obvious which line relates to which axis we will change the vertical Axis font colours to match the respective line colour.


Click the chart's left axis and use the Font colour icon on the Home Ribbon tab to colour it blue.




Repeat for the right Axis and colour it red.

To complete the chart we will edit the chart title to display the measures being plotted.

Click the chart title on the chart and click in the Formula Bar and press the = key and use your mouse to click cell C3 and then press the Enter key. This links the chart title to cell C3. See below




To move the chart to the Chart sheet first select the chart, then click the Design or Chart Design Ribbon tab and click the Move Chart icon on the far right of the Ribbon and click the second drop down box and choose the Chart sheet as per image below and click OK.



On the Chart sheet change the fill colour of cell D8 to blue and cell D11 to red. This becomes the legend.

Move and re-size the chart as per the image below and the interactive chart is ready to use.

Making selections in the Combo Box and the using the Scroll Bar will automatically update the chart. See example below.



Interactive charts are one example of where using Form controls can simplify making selections on Excel spreadsheets.